
Kapitel 1. Einführung in die Tabellierung

1 Einführung in die Tabellierung

In diesem Tutorial wird die Tabellierung mit GESStabs anhand eines einfachen Beispielprojekts
demonstriert. Dafür nutzen wir eine kleine Befragung zum Thema Internetnutzung, welche im
Wintersemester 2010/ 2011 an der Universität Hamburg bearbeitet und für diese Einführung mit
Zufallszahlen gefüllt wurde.

Zu Übungszwecken empfehlen wir, die hier beschriebenen Beispiele selbst auszuführen. Sie müssen
aber nicht alles abtippen, die Dateien stellen wir Ihnen im Archiv EXAMPLE.zip bereit.

Erste Schritte

Installieren Sie GESStabs bitte zunächst an Ihrem Arbeitsplatz.

1. Führen Sie das von uns zur Verfügung gestellte Programm aus und folgen Sie dem Installa-
tionsmenü. Standardmäßig wird GESStabs auf Ihrem Computer unter c://GESS installiert.
Von uns erhalten Sie einen (Test-)Lizenzkey. Zu dessen einmaliger Gültigkeits-Prüfung ist
eine Internetverbindung nötig.

2. Für die Arbeit mit den GESStabs-Skripten können Sie jeden beliebigen Texteditor nutzen.
Wir empfehlen den Notepad++ Editor: Dieser ist effizient in der Anwendung und wird von
uns mit einem Syntax-Highlighting unterstützt (.zip-Datei auf unserer Homepage).

3. Das Lesen, Verarbeiten und Schreiben von SPSS-Dateien erfordert eine SPSS-Input/Output-
Schnittstelle. Diese sog. DLL-Dateien stellen wir Ihnen bereit. Speichern Sie diese bitte
ebenfalls im c://GESS-Verzeichnis ab.

4. Sie können erstellte Tabellen (u.a.) als PostScript-Datei betrachten und als PDF-Dokument
abspeichern. Dazu benötigen Sie einen PostScript-Viewer sowie PDF-Destiller. Wir emp-
fehlen GhostView und GPL GhostScript, beides können Sie sich kostenlos im Internet
herunterladen.

5. Teilen Sie als letzten Schritt der GESStabs-Software unter Optionen die Verzeichnisse
mit, in denen auf die notwendigen Arbeitsprogramme (Texteditor, PostScript-Viewer und
PDF-Destiller) zugegriffen werden kann.

Im Folgenden werden Sie bei Ihren ersten Schritten mit GESStabs begleitet.
Nachdem Sie dieses Tutorial zusammen mit der beiliegenden Beispieldatei durchgearbeitet haben,
können Sie...

• ein Tabellierskript sinnvoll anlegen,

• mit einem ASCII-, SPSS- oder CSV-Datensatz arbeiten,

• Kreuztabellen (mit mehreren Köpfen) erstellen,

• Top(2)- und Bottom(2)-Boxes generieren,

• nach Häufigkeiten sortieren,

• recodieren,

• Mittelwerte und Mittelwertstabellen ausgeben und

• das Aussehen der Tabellen variieren.
Legen wir also los!

GESStabs ist ein Skript-orientiertes Tabellierprogramm. Das bedeutet: Alle Anweisungen an die
Software werden als Sammlung von Befehlen in Form eines Skripts festgehalten. Dieses Skript
wird anschließend von GESStabs abgearbeitet und erstellt dann die dort angewiesenen Tabellen.

Es ist guter GESS-Programmierstil, ein Skript in mehrere Dateien aufzuteilen. Unter unseren
Nutzern hat sich in den letzten 25 Jahren folgendes Vorgehen als effizient erwiesen:
Eine Datei mit der Extension *.tab wird als Hauptprogramm genutzt. Diese Datei bindet
mit dem INCLUDE-Statement weitere, thematisch definierte Dateien (Unterprogramme) ein.
Dieses Hauptprogramm sieht z. B. so aus:

Abb. 1.1: Das main.tab
In diesem Tutorial heißt das Hauptprogramm main.tab. Eingebunden werden in der gelisteten
Reihenfolge:

• die Datenquelle (...\data\test.dat),
• die Variablenbeschreibung des eingelesenen Datensatzes,
• das Tabellenlayout in der Datei tutorial.fmt,
• Variablenumformungen in der Datei compute.inc
• die Definition der Tabellenköpfe1 in der Datei kopf.inc
• und die eigentlichen Tabellieranweisungen mit dem Namen tab.inc.

GESStabs liest diese in unterschiedliche Dateien aufgeteilten Steueranweisungen von oben nach
unten, sodass letztendlich ein einziger Stapel als Abfolge aller einzelnen Anweisungen abgearbeitet
wird.

1In der Markt- und Meinungsforschung werden Kreuztabellen oft so dargestellt, dass oben in horizontaler Richtung
die sogenannten Aufbruchgruppen aufgeführt werden. Die interessierenden Fragen werden dann senkrecht, meist
in Form von Spaltenprozenten, dargestellt. Diese oben in der Tabelle stehenden Aufbruchgruppen werden in
GESS-Anleitungen als „Kopf“ bezeichnet.

Kapitel 1. Einführung in die Tabellierung

Die Steueranweisungen im main.tab enthalten neben dem Verweis auf die INLCLUDE-Dateien
folgende Befehle:

PRINTFILE PS = tutorial.ps;

AUTONOANSWER = YES "k.A.";
USEMISSING = NO;
USECASES = YVALID;
CITEFIRSTVAR = TOPTEXT YVALID;

SETMISSING = -1;
BLANKVALUE = -1;

DATAFILE = "..\daten\test.dat";
VARIABLE lfd = 31 30;
KEY OPENQFILE = lfd;
OPENQFILE = "..\daten\open.opn";

END;

Mit ein wenig Übung sind viele der Steueranweisungen selbsterklärend. Gehen wir die einzelnen
Zeilen einmal durch.

Ein Semikolon beendet jede einzelne Steueranweisung.

Mit PRINTFILE ps = tutorial.ps; wird eine Ausgabedatei benannt. Auf das Schlüsselwort
PRINTFILE folgen die Bezeichnung des Ausgabeformats, in diesem Fall PostScript (ps), und die
Dateibenennung, hier ’tutorial.ps’.

AUTONOANSWER = YES; bewirkt, dass fehlende Antworten standardmäßig und für jede Frage als
"k.A." gesetzt werden (Filter werden berücksichtigt)2. USEMISSING veranlasst, dass fehlende Werte
nicht in Berechnungen (z.B. Mittelwerte) eingehen. Mit der Anweisung USECASES = yvalid;
wird bestimmt, dass in der Tabellierung nur die gültigen Fälle des Seitenaufrisses, also der
Y-Variablen, berücksichtigt werden. CITEFIRSTVAR = toptext yvalid; heißt: Drucke den
Toptext (üblicherweise ist dies der Fragetext) der Vorspaltenvariable über der Tabelle ab.

Mit SETMISSING wird der interne Missing-Wert bestimmt, hier -1. BLANKVALUE gibt den Wert
an, der bei leeren Datenzellen gesetzt werden soll. Hier entspricht dieser dem Missing-Wert, d.h.
leere Datenfelder werden als Missing interpretiert.

Den Schlusspunkt des (aktiven) Skriptes setzt das Schlüsselwort end; . Alle danach folgenden
Texte werden zwar gelesen, jedoch nicht mehr interpretiert.

2 AUTONOANSWER = NO; wirkte basisreduzierend.

2 Einlesen von Datensätzen

Nachdem die generelle Systematik eines GESStabs-Skripts erklärt wurde, widmen wir uns nun
der nächsten grundsätzlichen Frage: Wie bindet man einen Datensatz ein?
In der Regel wird heutzutage mit ASCII-, SPSS- oder CSV-Datensätzen gearbeitet.
Der einzige Unterschied in der Handhabung der Datenformate liegt in der jeweiligen Variablen-
definition. Während in SPSS feste Variablenbeschreibungen gespeichert werden können, müssen
die Variablen aus ASCII- und CSV-Datensätzen erst noch definiert werden. SPSS- und CSV-
Datensätze ähneln sich strukturell, in ASCII wird die Variablen-Zuordnung über Spaltenangaben
gesteuert.
Es folgt eine kurze Einführung in die drei Varianten.

2.1 ASCII

ASCII-Datensätze werden wie folgt in das Skript eingebunden1:
DATAFILE = "Dateiname.dat";

In ASCII-Datensätzen entspricht eine Zeile einem Fall. Variablen werden anhand ihrer Spaltenzahl
beschrieben.

2.1.1 Manuelles Erstellen einer var.inc

Liegt noch keine vorbereitete var.inc vor2, muss diese selbst geschrieben werden.
Grundsätzlich braucht man für eine Variablendefinition nur den Variablentypen, den Namen und
die Spaltenangabe.

Singlenennungen

Die Definition einer Frage mit Singlenennung sieht beispielsweise so aus:
SingleQ f0 = 30 1;

Hiermit ist folgendes definiert:
Die Variable enthält maximal eine Antwort pro Interview, ist also eine Variable vom Typ
SingleQ. Die Variablenbezeichnung lautet f0, hiermit kann die Variable später aufgerufen und
weiter verarbeitet werden. Die Ziffern nach dem Gleichheitszeichen geben die Spaltenposition
im Datensatz und die Anzahl der ab dort genutzten Spalten an. Die Variable f0 befindet sich
demnach in der 30. Spalte des Datensatzes und ist eine Spalte breit.
Anschließend kann man der Frage einen TEXT und einen TITLE zuweisen.
TEXT = "Seit wann nutzen Sie das Internet?";
TITLE = "Seit...";

Liegt keine Zuweisung zu einer bestimmten Frage vor, werden diese der zuletzt definierten
Frage zugeordnet. Für eine direkte Zuweisung kann der Variablenname hinter das Befehlswort
geschrieben werden3.
1Siehe Erste Schritte für das Vorgehen bei zusätzlichem Einbezug eines OpenQ-Datenfiles.
2Für Umfragen, die mit GESS Q. oder GESS Cati durchgeführt wurden, lässt sich diese automatisch generieren,
siehe 2.1.2

3Mit diesem Vorgehen sind Sie auf der sicheren Seite, falls sich einmal die Reihenfolge des Skriptinhaltes ändert.

Der LABELS-Befehl bezeichnet die Inhalte der einzelnen Variablencodes.

LABELS f0 =
1 "weniger als 6 Monaten"
2 "6 - 12 Monaten"
3 "ein bis zwei Jahren"
4 "zwei bis drei Jahren"
5 "mehr als drei Jahren"
;

Multinennungen, Dichotome

Die Beschreibung von Variablen mit mehreren Antwortmöglichkeiten ähnelt dem Vorgehen für
SingleQ-Variablen.

MultiQ f9 = 392 22 2;
TEXT = "Für welche Zwecke nutzen Sie das Internet?";
TITLE = "Bitte klicken Sie alles zutreffende an.";
LABELS =
1 "Recherche/Ratgeber"
2 "Videos/ Musik"
3 "E-Mail"
4 "Online Gaming"
5 "Shopping / Booking"
6 "Ämter / Verwaltung (Uni, Steuer)"
7 "Chats & Foren"
8 "Online Banking"
9 "Online Medien (Zeitschriften, Magazine, Blogs)"
10 "Soziale Netzwerke (Facebook, Parship, XING)"
11 "Sonstiges"
;

Die Definition der Spaltenangabe ist hier ein wenig anders: Insgesamt benötigt eine MultiQ
genauso viele Spalten, wie Angaben gemacht werden können (Varibale f9 hat 11 Codes, kann
daher maximal 11 Antworten erhalten, daher hier: 11 Spalten). Weil die Labelcodes 10 und 11
zweistellig sind, muss jede mögliche Antwort Platz für zwei Ziffern bereitstellen. Dementsprechend
werden höchstens 11x2 = 22 Spalten benötigt. Oben in der Spaltenangabe ist spezifiziert, dass
die MultiQ f9 in Spalte 392 des Datensatzes beginnt, insgesamt 22 Spalten benötigt und jede
Angabe maximal zwei Stellen besitzt.

2.1.2 Automatisch generierte var.inc

Wurde eine Umfrage in GESS Q. oder GESS Cati durchgeführt, kann eine zugehörige var.inc
einfach per Mausklick erstellt werden. Hierin sind alle Variablen mit Spaltendefinition und -breite
enthalten. Die Vertextung der Variablen entspricht der in der Umfrage verwendeten Texte.

Für unseren Beispiel-Datensatz konnten wir eine automatisch generierte var.inc erstellen.

Die Variablendefinition von f1 sieht beispielsweise so aus:

SingleQ f1 = 478 1;
TEXT = "An welchem Ort nutzen Sie jetzt gerade das Internet?";
LABELS =
1 "zu Hause"
2 "Am Arbeits- / Ausbildungsplatz"
3 "In der Schule / an der Universität"
4 "An einem öffentlichen Ort (Internet-Cafe, Multimedia-Terminal)"
5 "Von unterwegs (über WLAN Hotspot, Handynetz, ...)"
6 "Woanders außer Haus (bei Freunden, Verwandten)"
;

In der zugehörigen Befragung war kein Title vorgesehen. Wird dieser im Zuge der Auswertung
gewünscht, kann er mit dem TITLE-Befehlsschlagwort ergänzt werden. Zum Beispiel:

TITLE f1 = "Ort der Internetnutzung";

Die f3, eine Frage mit Mehrfachantwortmöglichkeit, wurde analog beschrieben. Mit sieben Labels,
von denen eines zweistellig ist, benötigt die Variable maximal 7x2 = 14 Spalten je Fall. Text,
Title und Labels sind hier ebenfalls automatisch definiert.

MultiQ f3 = 479 14 2;
TEXT = "An welchen der folgenden Orte nutzen Sie zumindest gelegentlich das
Internet?";
TITLE = "Bitte kreuzen Sie alles an, was auf Sie zutrifft!";
LABELS =
1 "Zu Hause"
2 "Am Arbeits- / Ausbildungsplatz"
3 "In der Schule / an der Universität"
4 "An einem öffentlichen Ort (Internet-Cafe, Multimedia-Terminal)"
5 "Von unterwegs (über WLAN Hotspot, Handynetz, ...)"
6 "Woanders außer Haus (bei Freunden, Verwandten)"
99 "Keinen davon"
;

Die Beschreibung einer Matrix-Frage kann nicht einfach 1:1 aus dem Fragebogenskript übernom-
men. Hier wird stattdessen ein Makro erstellt und automatisch in die var.inc integriert.

#macro #f10.1(&gItemVal &title &col)
SingleQ f10.&gItemVal.1 = &col 1;
TEXT = "Wie häufig nutzen Sie diese Angebote im Internet?";
TITLE = "&title";
LABELS =
1 "sehr häufig"
2 "häufig"
3 "gelegentlich"
4 "selten"
5 "sehr selten"
;
#endmacro

#f10.1(1 "Recherche/Ratgeber" 520)
#f10.1(2 "Videos/Musik" *)

#f10.1(3 "E-Mail" *)
#f10.1(4 "Online Gaming" *)
#f10.1(5 "Shopping/Booking" *)
#f10.1(6 "Ämter/Verwaltung (Uni, Steuer)" *)
#f10.1(7 "Chats & Foren" *)
#f10.1(8 "Online Banking" *)
#f10.1(9 "Online Medien (Zeitschriften, Magazine, Blogs)" *)
#f10.1(10 "soziale Netzwerke (Facebook, Parship, XING)" *)

Im ersten Abschnitt ist die Fragestruktur definiert. Für die Itemwerte, Title und Angabe der
Startspalte werden Platzhalter gesetzt. Im zweiten Abschnitt wird das Makro für jedes Frageitem
aufgerufen, das jeweils als eigene Variable abgelegt ist. Werte für Itemwert, Title und Startspalte
werden eingesetzt. Ab dem zweiten Makroaufruf ist ein Sternchen anstelle der Startspalte gesetzt.
Dies bewirkt ein einfaches Hochzählen der Spaltenangabe: Item 1 liegt auf Spalte 520, für Item 2
wird mit 521 hochgezählt, Item 3 liegt in Spalte 522, usw..

2.2 SPSS

SPSS-Datensätze werden mit dem Schlagwort SPSSINFILE in das Skript einbezogen:

SPSSINFILE = "Dateiname.sav";

Auch SPSS-Datensätze führen einen Fall je Zeile auf. Das Programm kann Variablentexte und
-labels speichern. Im Idealfall arbeiten Sie mit einem SPSS-Datensatz, der schön mit Text- und
Labeltexten bestückt ist, und können direkt mit der Tabellierung beginnen. Dies ist leider selten
der Fall.

Überblick über einen SPSS-Datensatz

GESStabs stellt verschiedene Möglichkeiten bereit, um sich einen Überblick über einen gegebenen
Datensatz zu machen.

Mit dem Schlagwort SYNTAX können verschiedene Variableninformationen aus dem Datensatz in
Textdateien herausgeschrieben werden, nämlich die Variablentypen, -texte, -title und -labels. Zur
Extraktion der Eigenschaften aus dem Datensatz wird das jeweilige Stichwort dem SYNTAX-Befehl
nachgestellt.
Es ist Geschmackssache, ob alle Informationen in eine einzige Textdatei geschrieben werden:

SYNTAX VARIABLES TITLE TEXT LABELS = alles_zusammen.inc;

oder thematisch aufgeteilt werden:

SYNTAX VARIABLES = variablen.inc;
SYNTAX TITLE = title.inc;
SYNTAX TEXT = texte.inc;
SYNTAX LABELS = labels.inc;

An dieser Stelle sei erwähnt: SPSS unterscheidet nicht zwischen Texten und Titlen, führt
stattdessen einfache Variablenlabels. GESStabs weist diese standardmäßig als Variablentitle aus.

Die erstellten Dateien enthalten nun die Informationen über Typen, Title und Labels der im
Datensatz enthaltenen Variablen:

Abb. 2.1: Variablenbezeichnungen aus dem SPSS Datensatz

Abb. 2.2: Vartitle aus dem SPSS Datensatz

Eine andere Möglichkeit, sich einen Überblick über einen gegebenen SPSS-Datensatz zu machen,
bietet das CODEBOOK-Statement .

Dies bewirkt eine einfache Häufigkeitsauszählung über alle

CODEBOOK;

oder ausgewählte Variablen

CODEBOOK f1;

und sieht z.B. so aus:

Abb. 2.3: Häufigkeitsauszählung
Tipp: Codebooks eigenen sich auch wunderbar, um Ihre anschließend erstellten Tabellen auf
Datenrichtigkeit zu kontrollieren.

2.3 CSV

Mit CSVINFILE werden CSV-Datensätze in ein GESStabs-Projekt eingebunden:

CSVINFILE = "Dateiname.csv";

Strukturell ähneln CSV-Datensätze SPSS-Files: Jede Zeile enthält einen Interviewfall.

CSV-Datensätze speichern - neben den Rohdaten und Variablennamen - allerdings keine Va-
riableninformationen. Diese können Sie nachträglich definieren, siehe dafür Manuelles Erstellen
einer var.inc.
Um die korrekte Darstellung offener Texte (und insbesondere enthaltener Umlaute) im Tabellen-
Output sicherzustellen, empfehlen wir, die Kodierung der CSV-Datendatei standardmäßig auf
UTF-8 einzustellen:

ENCODING CSV = UTF8;

3 Einfache Kreuztabelle aus bestehenden
Variablen

Das Einfachste ist, zwei bestehende Variablen in einer Kreuztabelle gegeneinander zu tabellieren.
Hierfür definieren wir eine Tabelle, die eine Variable im Kopf und eine im Seitenaufriss enthält.
Wir möchten die Verteilung der Frage f1 (Ort der Internetnutzung) nach Geschlecht (Variable
s3) betrachten und schreiben dazu folgendes Statement in die tab.inc:

TABLE = s3 BY f1;

Das ist die einfachste Form eines GESStabs-Tabellen-Statements. Auf das Schlüsselwort TABLE
und ein Gleichheitszeichen folgt die Kopfvariable. Hinter BY wird die Variable geschrieben, die
in der Vorspalte1 stehen soll. Letztere ist die eigentlich interessierende Variable, deren Verteilung
betrachtet wird. Abgeschlossen wird die Anweisung, wie immer, durch das Semikolon.

Zum Tabellieren rufen wir das GESStabs-Menü auf, es öffnet sich folgende minimalistische
Programm-Oberfläche:

Abb. 3.1: Die GESStabs-Oberfläche

1Die Bezeichnungen hierfür variieren: Vorspalte, Aufriss, Y-Variable...

Über den Schalter select tab-file and run rufen wir unser main.tab auf. GESStabs führt das
Skript aus und zeigt folgende Meldungen an:

Abb. 3.2: Erstellen von Tabellen
Unser Beispiel-Programm wurde erfolgreich abgearbeitet, rechts oben kann die Anzahl verarbei-
teter Skriptzeilen, einbezogener Datenfälle und erstellter Tabellen nachvollzogen werden.
Ein Klick auf den Button view PostScript öffnet folgende Tabelle im benutzten PostScript-
Viewer:

Abb. 3.3: einfache Kreuztabelle

Das Ergebnis ist eine einfache Kreuztabelle. Über der Tabelle wird der Text der Vorspalten-
Variable abgebildet, über den Labels der beiden Variablen steht der jeweilige Title. Die Tabelle
zeigt absolute Werte (siehe Beschriftung ’Abs.’) und enthält eine Totalspalte sowie -zeile.
Ein Klick auf make PDF würde eine zusätzliche, zur PostScript-Ansicht identische PDF-Datei
erzeugen. Für diesen Moment belassen wir es aber bei der PostScript-Datei und lassen diese

geöffnet. Nach erneutem Start des GESStabs-Laufs (Klick auf run again ) aktualisiert sich diese
automatisch, wodurch ein schnelles betrachten vorgenommener Änderungen und Ergänzungen
möglich ist.

4 Kreuztabelle mit mehreren Variablen im Kopf

Eine Tabelle mit lediglich zwei Spalten im Kopf sieht nun etwas langweilig aus. Wir fügen dem
Tabellenkopf daher weitere Variablen hinzu. Dafür ergänzen wir einfach die entsprechenden
Variablennamen im TABLE-Statement:

TABLE = s1 s2 s3 BY f1;

Diese Variablen (Gerätenutzung, Hauptwohnsitze & Geschlecht) sollen also unseren neuen Kopf
bilden.
Für Tabellenköpfe, die häufig benutzt werden, ist es unpraktisch, diese in jeder neuen Tabellen-
definition mit allen enthaltenen Variablennamen erneut aufzuführen. Zu diesem Zwecke bietet
sich ein universaler Mechanismus der GESStabs-Skriptsprache an: das #EXPAND. Dabei handelt
es sich um einen einfachen Textersatz.

Wir nennen unseren Tabellenkopf #K11 und definieren mit folgendem Statement alle Variablen,
die enthalten sein sollen:

#EXPAND #K1 s1 s2 s3

Als einfacher Textersatz wird anschließend alles übertragen, was in der Zeile auf das Schlüsselwort
#EXPAND folgt, inkl. Kommentarzeichen, Semikolon etc.. Hier darf also kein abschließendes
Semikolon stehen, da dieses den TABLE-Befehl Hinter ’#K1’ beenden würde.

Die Tabellenanforderung für die identische Tabelle verkürzt sich dann auf:

TABLE = #K1 BY f1;

Jetzt kann man fortfahren und die gleiche Frage mit anderen Köpfen tabellieren. Oft möchte
man mehr Variablen gegeneinander betrachten, als auf eine Tabellenseite passt. Daher erstellen
wir einen zweiten Tabellenkopf, der Fragen nach Einkommen und Bildung enthält.

#EXPAND #K2 s8 s9

Effektives Skripting erfordert Übersichtlichkeit. Deshalb werden Tabellenköpfe von uns in einer
eigenen Datei angelegt, in der eingangs erwähnten kopf.inc. In unserem Beispiel enthält diese
Datei nur diese zwei Zeilen:

#EXPAND #K1 s1 s3 s2
#EXPAND #K2 s8 s9

Da unser Beispiel noch recht übersichtlich (weil kurz) ist, mag diese Aufteilung zu streng
erscheinen. In der Praxis wird jedoch häufig mit zahlreichen und komplexen Tabellenköpfen
gearbeitet. Andere Programmierer (oder man selbst, ein halbes Jahr später) finden sich leichter
zurecht, wenn man sich an solche Formalismen hält.

Anschließend tabellieren wir unsere Frage f1 mit beiden Köpfen:

TABLE = #K1 BY f1;
TABLE = #K2 BY f1;

1Namen von EXPANDs müssen stets mit einem Lattenkreuz beginnen.

Als Ergebnis erhalten wir die folgenden beiden Tabellen:

Abb. 4.1: Tabellenkopf mit Textersatz

Abb. 4.2: Kreuztabelle mit mehreren Köpfen

5 Tabellen bearbeiten

5.1 Groups

Der zweite Kopf ist nun etwas lang und die Tabelle dadurch nicht sehr leserlich, was hauptsächlich
an den zahlreichen Labels aus Variable s9 liegt. Es bietet sich an, die Einkommensklassen im
Nachhinein weiter zusammenzufassen1. Dafür generieren wir eine neue Variable, genauer gesagt
eine GROUPS. Mit einer GROUPS können Labels zusammengefasst, neue Labeltexte und -title
vergeben und/ oder Variablen gefiltert werden.

GROUPS s9neu =
| "Kein eigenes Einkommen" : 10 in s9
| "1 bis unter 1.000 " : [1:2] in s9
| "1.000 bis unter 2.500 " : [3:5] in s9
| "2.500 bis unter 3.500 " : [6:7] in s9
| "3.500 und mehr" : [8:9] in s9
;
TITLE s9neu = "Einkommen";

Die Labeldefinitionen einer GROUPS beginnen immer mit einem |, gefolgt vom Labeltext. Nach
dem Doppelpunkt wird definiert, welche Werte aus s9 in das jeweilige Label einfließen sollen2.
Unter Einbezug der neu generierten GROUPS wird nun ein neuer Tabellenkopf generiert:

#EXPAND #K3 s8 s9neu

Tabelliert sieht dies so aus:

TABLE = #K3 BY f1;

Abb. 5.1: Anpassen der Kopf-Variablen

5.2 Overcodes

Mithilfe des LABELS-Statements können einer bestehenden Variable spielend leicht Labels hinzu-
gefügt, diese sortiert und bearbeitet werden.

Schauen wir uns einmal die Variable f11_1 an:

1Unserer Skript-Systematik folgend nehmen wir Variablenmanipulationen in der compute.inc vor.
2Achten Sie darauf, die Labels sauber zu definieren. Überschneiden sich die Bedingungen, können Werte eines
Falls in mehrere Labels der Groups einfließen.

SingleQ f11_1 = 713 2;
TEXT f11_1 = "Und zu welchem Thema recherchieren Sie am häufigsten?";
TITLE = "Recherche";
LABELS f11_1 =
1 "Nachrichten"
2 "Familie und Kinder"
3 "Wirtschaft/ Finanzen"
4 "Politik"
5 "Sport"
6 "Essen und Trinken"
7 "Erotik"
8 "Kino- und Fernsehprogramm"
9 "Wetter"
10 "Beauty und Fashion"
11 "Freunde/ Partnerschaft"
12 "Kunst/ Kultur/ Musik"
;

Die Themen sind hier ziemlich durcheinandergewürfelt und nicht nach Überpunkten geordnet.
Mit GESStabs können wir die Reihenfolge der Labels einfach selbst bestimmen.
LABELS f11_1 =
1 "Nachrichten"
4 "Politik"
9 "Wetter"
3 "Wirtschaft/ Finanzen"

10 "Beauty und Fashion"
6 "Essen und Trinken"
8 "Kino- und Fernsehprogramm"
12 "Kunst/ Kultur/ Musik"
5 "Sport"

7 "Erotik"
2 "Familie und Kinder"
11 "Freunde/ Partnerschaft"
;

Im Ausdruck einer Tabelle würden jetzt die Themen in dieser Reihenfolge erscheinen. Nun ist es
häufig wünschenswert, Oberthemen zu bilden und diese auszählen zu lassen. Hierfür muss man
den Labels lediglich die übergeordneten Begriffe als OVERCODE mitteilen.
LABELS f11_1 =
OVERCODE 1 3 4 9 "Öffentlich (Nettozählung)"
1 "Nachrichten"
4 "Politik"
9 "Wetter"
3 "Wirtschaft/ Finanzen"
OVERCODE 5 6 8 10 12 "\Lifestyle (Nettozählung)"
10 "Beauty und Fashion"
6 "Essen und Trinken"
8 "Kino- und Fernsehprogramm"
12 "Kunst/ Kultur/ Musik"

5 "Sport"
OVERCODE 2 7 11 "\Privat (Nettozählung)"
7 "Erotik"
2 "Familie und Kinder"
11 "Freunde/ Partnerschaft"
;

Würde das Schlüsselwort OVERCODE um das Schlüsselwort SUM ergänzt, würde eine Brutto-Zählung
vorgenommen.

Mit #K1 im Kopf sieht die Tabelle der neuen f11_1 so aus:

Abb. 5.2: Overcodes

5.3 Top(2) und Bottom(2) Boxes

Oft sollen Skalen neben der Häufigkeitsverteilung auch in Form von Top- oder Bottom-Boxes
abgebildet werden. Auch dies kann einfach mithilfe von OVERCODEs umgesetzt werden.
Hier ein Beispiel für die Formulierung von Overcodes in Frage 10.1.1:

SingleQ f10.1.1 = 520 1;
TEXT f10.1.1 = "Wie häufig nutzen Sie diese Angebote im Internet?";
TITLE f10.1.1 = "Recherche/Ratgeber";
LABELS f10.1.1 =
1 "sehr häufig"
2 "häufig"
3 "gelegentlich"
4 "selten"
5 "sehr selten"
Overcode 1 2 "Top-2-Box (1+2)"
Overcode 4 5 "Bottom-2-Box (4+5)"
;

... und dessen tabellarische Darstellung:

Abb. 5.3: Tabelle mit Overcodes

5.4 Vargroup

Datensätze enthalten immer wieder Itembatterien, deren Top- und Bottom-Boxes im Tabellenband
zusammengefasst dargestellt werden sollen. In GESStabs gibt es den Variablentyp VARGROUP, der
hierfür gut geeignet ist.

Eine VARGROUP setzt sich aus mehreren einfachen Variablen zusammen. Eine VARGROUP aus n
Variablen hat automatisch n Label, die nach definierten Bedingungen berechnet werden.

Das ist in der Praxis einfacher als in der Theorie. Nehmen wir wieder eine Matrix-Variable aus
unserem Beispieldatensatz. Diese besteht aus zehn Items, die die Nutzungshäufigkeit bestimmter
Internetangebote erfragen (f10). Definiert ist diese Itembatterie wie folgt:

#macro #f10.1(&gItemVal &title &col)
SingleQ f10.&gItemVal.1 = &col 1;
TEXT = "Wie häufig nutzen Sie diese Angebote im Internet?";
TITLE = "&title";
LABELS =
1 "sehr häufig"
2 "häufig"
3 "gelegentlich"
4 "selten"
5 "sehr selten"
;
#endmacro

#f10.1(1 "Recherche/Ratgeber" 520)
#f10.1(2 "Videos/Musik" *)
#f10.1(3 "E-Mail" *)
#f10.1(4 "Online Gaming" *)
#f10.1(5 "Shopping/Booking" *)
#f10.1(6 "Ämter/Verwaltung (Uni, Steuer)" *)
#f10.1(7 "Chats & Foren" *)
#f10.1(8 "Online Banking" *)
#f10.1(9 "Online Medien (Zeitschriften, Magazine, Blogs)" *)
#f10.1(10 "soziale Netzwerke (Facebook, Parship, XING)" *)

Wir möchten nun eine Übersicht der Top-2- und Bottom-2-Boxes aller Items erstellen. Dazu
zählen wir alle Einzelvariablen (diese entsprechen den einzelnen Items des Fragekomplexes f10)
auf, die einbezogen werden sollen, und definieren anschließend die Berechnung der Top-2- und
Bottom-2-Boxes:

VARGROUP top2 = (f10.1.1 f10.2.1 f10.3.1 f10.4.1 f10.5.1 f10.6.1 f10.7.1 f10
.8.1 f10.9.1 f10.10.1) eq 1 2;
VARTITLE top2 = "Top-2-Box \sehr häufig + häufig";

VARGROUP bot2 = (f10.1.1 f10.2.1 f10.3.1 f10.4.1 f10.5.1 f10.6.1 f10.7.1 f10
.8.1 f10.9.1 f10.10.1) eq 4 5;
VARTITLE bot2 = "Bottom-2-Box \sehr selten + selten";

Eine VARGROUP setzt automatisch die ursprünglichen TITLE der einfachen Variablen als LABELS
der neu generierten Vargroup um.

Tabelliert sieht das Ganze schließlich so aus:

TABLE = #K1 BY top2;
TABLE = #K1 BY bot2;

Abb. 5.4: Top2-/ Bottom2-Boxes

5.5 Mittelwerte

Sobald eine Frage eine Skala enthält, möchte man in der Regel auch einen Mittelwert ausweisen.
Das kann in GESStabs denkbar einfach, direkt im Zuge einer Tabellieranweisung erledigt werden.

Als Beispiel ziehen wir noch einmal die Variable f10.1.1 heran3.
Einen Mittelwert fordern wir im TABLE-Statement mit dem Schlüsselwort MEAN und der Spezifi-
kation der gewünschten Variable an:
3Siehe Top(2) und Bottom(2) Boxes.

TABLE = #K1 BY f10.1.1 MEAN(f10.1.1);

Hinweis: Es empfiehlt sich, Mittelwerte nur auf Basis der gültigen Antworten zu berechnen und
(häufig mit -1 oder 99 vercodete) fehlende Antworten aus der Berechnung auszuschließen. Letztere
würden den Mittelwert irrtümlich verfälschen. Dies haben wir in den ersten Zeilen unseres main.
tab festgelegt. Alle Befragte, welche keine Antwort gegeben haben, sollen automatisch "k.A."
erhalten und Missings Values nicht in die Berechnung einfließen: AUTONOANSWER = YES "{k.A.";
USEMISSING = no;.

Abb. 5.5: Mittelwerte

In der Zeile, die die Mittelwerte beschreibt, steht nun noch der Variablentitle von f10.1.1. Damit
dort korrekterweise "Mittelwert" als Zeilenbeschreibung erscheint, erweitern wir die Tabellier-
Anweisung um das DESCRIPTION-Stichwort:

TABLE = #K1 BY f10.1.1 MEAN :DESCRIPTION "Mittelwert"(f10.1.1);

Abb. 5.6: Mittelwerte

Dies könnte nun für alle folgenden Items des Fragekomplexes fortgeführt werden. Als Ergebnis
erhielten wir zehn Tabellenseiten, auf denen jeweils die Skalen-Verteilung plus Mittelwert der
einzelnen Items aufgeführt wären. Nicht sehr übersichtlich, oder?
Um die Items auf einen Blick vergleichen zu können, bietet es sich stattdessen an, eine Mittelwerte-
Übersicht zu erstellen. Das ist ganz einfach:

TABLE = #K1 BY MEAN(f10.1.1) MEAN(f10.2.1) MEAN(f10.3.1) MEAN(f10.4.1) MEAN(f10
.5.1) MEAN(f10.6.1) MEAN(f10.7.1) MEAN(f10.8.1) MEAN(f10.9.1) MEAN(f10.10.1);

Hier haben wir die :DESCRIPTION-Anweisung wieder herausgenommen und nutzen so den bereits
vergebenen Title der einzelnen Items. Hier die Tabelle:

Abb. 5.7: Mittelwerte-Übersicht

5.6 Sortieren nach Häufigkeiten

Ebenfalls der Übersichtlichkeit halber bietet es sich häufig an, eine tabellierte Variable nach
Häufigkeit ihrer Nennungen zu sortieren. Beispielhaft skripten wir dies für Frage f9 aus unserem
Datensatz.
Für eine absteigende Sortierung der absoluten Werte (des Totals) muss die bekannte Tabellie-
rungsanweisung nur um drei Schlüsselwörter ergänzt werden:

TABLE = #K1 BY f9 SORT ABSOLUTE DESCEND;

Abb. 5.8: Absteigend sortierte Tabelle

5.7 Recode

Häufig enthalten Umfragen auch numerische Fragen. In der Tabellierung macht es meist keinen
Sinn, sämtliche Einzelnennungen aufzuführen. Stattdessen können die Einzeleingaben der Be-
fragten für die Auswertung in Kategorien zusammengefasst werden4. Zu diesem Zwecke steht in
GESStabs der Befehl RECODE bereit. Ein RECODE erlaubt die Umkodierung einzelner Variablen-
ausprägungen der zuletzt definierten Variable oder einer Liste explizit benannter Variablen.
4Hierbei muss natürlich beachtet werden, dass eine Gruppierung die ursprüngliche Verteilung der Daten nicht
verzerrt.

Als Beispiel nehmen wir die Frage f7.1. Diese ist wie folgt in unser automatisch generierten
var.inc beschrieben:

#macro #f7(&p1, &title, &col, &width)
SingleQ f7&p1 = &col &width;
TEXT = "Wie viel Zeit haben Sie gestern aktiv im Internet verbracht?";
TITLE = "&title";
#endmacro

#f7(1 "Stunden:" 493 4)

Als Eingabe sind ganze Zahlen zwischen 0 und 24 (Stunden) erlaubt gewesen. Ohne einen RECODE
wird jede gemachte Eingabe abgebildet:

Abb. 5.9: vor dem Recode
Um die Einzel-Werte zu gruppieren, erstellen wir erst einmal eine neue Variable, genauer gesagt
eine identische Kopie der f7.1. Natürlich könnten wir auch direkt die ursprüngliche Variable
bearbeiten, jedoch gingen dann die Originalwerte verloren5. Es ist sicherer – und nicht aufwändiger
–, eine neue Variable zu generieren.

COMPUTE f7r = f7.1;
COPYTEXT f7r = f7.1;
COPYTITLE f7r = f7.1;

Mit dem Befehl COMPUTE wird eine neue Variable erstellt (f7r) in welche die ursprünglichen Werte
(aus f7.1) kopiert werden. COPYTEXT und COPYTITLE kopieren die Texte der Ausgangsvariable
und weisen sie der neuen Variable zu.
Um nun Gruppen zusammenzufassen, weisen wir die Ausgangswerte (0 bis 24) neuen Labels zu:

RECODE f7r
0:5 = 1/
6:10 = 2/
11:15 = 3/
16:20 = 4/
21:24 = 5
;

5Außerdem werden wir die Originalwerte noch benötigen.

GESStabs (Stand: April 2018) 22



Kapitel 5. Tabellen bearbeiten

Anschließend vertexten wir die neuen Labelcodes:

LABELS f7r =
1 "0 bis 5 Stunden"
2 "6 bis 10 Stunden"
3 "11 bis 15 Stunden"
4 "16 bis 20 Stunden"
5 "21 bis 24 Stunden"
;

Tabelliert sieht die Verteilung der neuen Variable so aus:

TABLE = #K1 BY f7r MEAN(f7.1);

Abb. 5.10: Mit Recode
Noch einen Kommentar zum TABLE-Statement: Den Mittelwert berechnen wir natürlich auf Basis
aller Einzelnennungen aus der Ursprungsvariable. Ein Mittelwerte auf Grundlage der gruppierten
f7r würde nur die Labelcodes zwischen 1 und 5 zur Mittelwerteberechnungen heranziehen –
mathematisch richtig, inhaltlich falsch.

6 Tabellenlayout

Nachdem wir uns den Grundlagen des Tabelleninhaltes gewidmet haben, kommen wir nun zu
einem ebenso wichtigen Teil der Auswertung, nämlich der Gestaltung des Tabellenbandes und
der einzelnen Tabellen.
Von der Schriftfarbe und -größe über die Farbgestaltung sowie das Einfügen von Logos und
Bildern, der Positionierung der Tabelle auf dem Blatt sowie Einstellung einzelner Abstände kann
mit der GESStabs-Syntax das gesamte Tabellenlayout individuell eingestellt werden. An dieser
Stelle können wir lediglich auf die Grundlagen des Tabellendesigns eingehen. Möchten Sie das
Layout Ihres Tabellenbandes neu aufsetzen, stehen wir Ihnen natürlich gern unterstützend zur
Seite!

6.1 Tabellen-übergreifende Formatierung

Grundsätzlich bündeln wir tabellenübergreifende Formatierungsanweisungen in einer separaten
Format-Datei, hier der tutorial.fmt. Alle Anweisungen gelten so lange, bis sie neu definiert
werden1.

Zu den ganz grundlegenden Formatierungsanweisungen zählen die folgenden:
NUMBERCHAR = #;
SPLITCHAR = ~;
SPLITCHARSTAY = -;
LINEFEEDCHAR = \;

NUMBERCHAR gilt als numerischer Platzhalter (s.u.). Wird dieser z.B. in Seitenzahlen verwendet,
wird automatisch hochgezählt.
SPLITCHAR und SPLITCHARSTAY legen Trennzeichen fest: Während letzteres einen dauerhaften
Umbruch markiert, stellt SPLITCHAR ein bedingtes Trennzeichen dar und markiert die Stelle, an
der umgebrochen wird, wenn es notwendig ist2.
LINEFEEDCHAR erzwingt einen Zeilenumbruch.

Fundamental ist auch USEFONT: Hiermit werden Schriftfarbe und -größe festgelegt. Für einen
reibungslosen GESStabs-Lauf müssen wir dies einmal generell festlegen:
USEFONT = "Helvetica" SIZE 9;

Zudem kann USEFONT für einzelne Tabellenelemente definiert werden, z.B. USEFONT ABSOLUTE =
"Helvetica" SIZE 8;, USEFONT MEAN = "Helvetica-Bold" SIZE 10; usw..

Mit FORMAT werden die Zahlenformate der unterschiedlichen Maßzahlen bestimmt. Wünscht man
etwa generell Mittelwerte mit zwei Nachkommastellen, schreibt man:
FORMAT MEAN = '#,##';

ALIGN, DISTANCE oder FRAMECOLOR sind weitere Schlagworte, mit denen beispielsweise Boxenin-
halte ausgerichtet, Abstände definiert oder Rahmenfarben festgelegt werden. Eine vollständige
Formatierungsdatei steht in unserem Beispielprojekt zur Verfügung.
1An dieser Stelle sei noch einmal auf die lineare Abarbeitung des Skiptes durch die Software hingewiesen:
Formatierungsbefehle gelten für alle nachfolgenden Skriptzeilen und so lange, bis der Formatierungsbefehl neu
definiert wird. Dies gilt dann wieder für alle nachfolgenden Skiptteile usw..

2An dieser Stelle wird dann das unter SPLITCHARSTAY definierte Zeichen eingesetzt.

6.2 Tabellengestaltung

Eine Tabelle setzt sich aus verschiedenen Bereichen zusammen. Hier eine Übersicht über die
Struktur einer typischen GESStabs-Tabelle:

Abb. 6.1: Übersicht Tabellengestaltung

HEADER und FOOTER werden üblicherweise für den ganzen Tabellenband festgelegt, in letzterem
steht üblicherweise die Seitennummer:

FOOTER = "-#-";

Auch die Positionen INSTITUTION und DOCUMENT ändern sich in der Regel nicht innerhalb
eines Tabellenbandes. Hier können beispielsweise der Institutsname und eine beliebige weitere
Anmerkung platziert werden:

INSTITUTION = "GESS Gesellschaft für Software in der Sozialforschung mbH";
DOCUMENT = "based on randomized numbers \ "DATE;

Letzteres haben wir mit DATE ergänzt – hier wird das jeweils aktuelle Datum der Tabellengene-
rierung eingesetzt.

Im TABLETITLE stehen meist der Name der Studie und/ oder eine thematische Zuordnung. Etwa:

TABLETITLE = "Demostudie: GESStabs-Tutorial";

Darunter steht der TOPTEXT der Vorspalten-Variable3

Linksstehend findet man eine Beschreibung der Aufrissvariable: gegebenfalls ist hier der TITLE
angezeigt und die Labelcodes- oder (wenn vergeben) -texte werden aufgelistet. Nach demselben
Schema wird in horizontaler Richtung die Kopfvariable aufgeführt.
3Theoretisch kann hier auch etwas anderes stehen, siehe die entsprechende Einstellung in Das main.tab.

In welcher Form die Daten innerhalb der Tabelle präsentiert werden, kann individuell gesteuert
werden.
Den Rahmen bilden die sogenannten FRAMEELEMENTS. In unserem Schema gibt es eine Totalzeile
und -spalte. Diese Elemente sind nicht automatisch vorhanden, sondern werden mit folgendem
Befehl hinzugefügt:

FRAMEELEMENTS = TOTALROW TOTALCOLUMN;

Was innerhalb der einzelnen Datenzelle stehen soll, welche Maßzahlen also für die abgebildeten
Daten ausgewiesen werden sollen, steuert des Befehlsstichwort CELLELEMENTS. Spaltenprozente
und Absolutwerte werden etwa wie folgt angefordert:

CELLELEMENTS = COLUMNPERCENT ABSOLUTE;

Wie im vorherigen Abschnitt 6.1 beschrieben, wird die Formatierung der einzelnen Maßzahlen
mit dem FORMAT-Schlagwort bestimmt, zum Beispiel:

FORMAT COLUMNPERCENT = "`#,# \%"';

Schließlich können wir noch steuern, wie die Zelleninhalte beschrieben werden. Dies erscheint
dann im oberen linken Tabellenbereich, unter dem Fragetext. Zum Beispiel:

DESCRIPTION COLUMNPERCENT = "Col \%";
DESCRIPTION ABSOLUTE = "Abs.";

Stichwortverzeichnis

Align, 24
ASCII, 2, 4

Multinennungen, 5
Singlenennungen, 4
var.inc, 4, 5, 22

Autonoanswer, 3

Blankvalue, 3
By, 10

Cellelements, 26
Citefirstvar, 3
Codebook, 8
Compute, 22
compute.inc, 2, 15
Copytext, 22
Copytitle, 22
CSV, 4, 9
CSVINFILE, 9

Date, 25
Description, 20, 21
Distance, 24
Document, 25

Encoding, 9
end;, 3
Erste Schritte, 1
Expand, 13

Footer, 25
Format, 24, 26
Formatierung, 24
Framecolor, 24
Frameelements, 26

GESStabs-Oberfläche, 10

Häufigkeitssortierung, 21
Header, 25

Include, 2
Institution, 25

kopf.inc, 2, 13
Kopfvariable, 13
Kreuztabelle, 10

Labels, 5, 8, 15

Layout, 24
Linefeedchar, 24

main.tab, 2, 11, 20
Make PDF, 11
Makro, 6
Mean, 19
Missing, 3, 20
Mittelwerte, 19
Mittelwerte-Übersicht, 20
MultiQ, 5

Notepad++, 1
Numberchar, 24

Overcode, 15, 17

PDF, 1, 11
PostScript, 1, 3, 11
Printfile, 3

Recode, 21
run again, 12

Select tab-file and run, 11
Setmissing, 3
SingleQ, 4
Sort, 21
Splitchar, 24
Splitcharstay, 24
SPSS, 1, 4, 7
SPSSINFILE, 7
Syntax, 7

tab.inc, 2, 10
Tabellengestaltung, 25
Table, 10, 13, 19
Tabletitle, 25
Text, 4
Title, 4, 6, 8
Top-, Bottom-Boxes, 17, 18
Toptext, 25

Usecases, 3
Usefont, 24
Usemissing, 3
UTF-8, 9

Vargroup, 18

